EURISOL DS/TASK12/TN-05-03
14 July 2005 (revised 13th October 2005)

The EURISOL Beta-beam Facility

Parameter and Intensity Values, Version 2

M. Benedikt, A. Fabich, S. Hancock, M. Lindroos

Abstract

An initial "bottom-up" analysis [1] of ion intensities along the accelerator chain is revised to take into account more recent simulations of the stacking of ^{18}Ne ions in the decay ring and beneficial trends [2] in output flux as functions of certain machine parameters. In addition, space charge detuning at injection in the PS has led to a rethink of the top energy of the RCS, while that at injection in the SPS has had an impact on the number of bunches per batch delivered by the PS. We present transverse emittance values (which enter the space charge tune shift calculations) together with an updated list of intensities for both ion species under consideration in the baseline scenario.

Stacking Ne ions

At given relativistic γ, the aperture of the decay ring defines a longitudinal acceptance limit which scales with ion momentum. Consequently, the more advantageous charge-to-mass ratio of ^{18}Ne ions should allow the number of merging steps to fill the decay ring to be increased by something approaching a factor of 3 with respect to the stacking procedure [3] established for ^{6}He involving 15 merges. However, the sensitivity of the process to phase errors between the two rf components employed in the merging puts an upper limit of about 20 on the number of merges that can realistically be achieved. We therefore assume that the extra acceptance available for ^{18}Ne in the upstream PS and SPS is exploited to stabilize the beam and consider 2 eVs per bunch (cf., 1 eVs for ^{6}He) injected into the decay ring, giving 40 eVs in the stack.

The resultant relative momentum spread of the stack is $\pm 2.5\times10^{-3}$ at the start of merging for both ion species and, applying the full 20+20 MV of the 40+80 MHz rf systems in the decay ring, the duration of each bunch is 5.2 and 4.5 ns for ^{6}He and ^{18}Ne, respectively.

Trends

Following one trend identified in [2], we choose to increase the accumulation time at the ECR by reducing the repetition rate of the RCS to 10 Hz (cf., 16 Hz in [1]). And, since the lifetime at injection in the PS is still much longer than the cycle time of the RCS, it likewise pays to increase the number of bunches injected. Consequently, we take the maximum rf harmonic consistent with the upper frequency limit of the cavities in the PS, viz., h=21 for both ion species, so that the number of bunches from the RCS can be increased to 20 (cf., 16 in [1]). This leaves one rf bucket empty to accommodate

the extraction kicker rise time and, even at the lower repetition rate of the RCS, incurs no penalty in PS cycle time due to its quantization.

We no longer assume a merging to take place in the PS as this would double the tune shift in the downstream SPS. The absence of PS merging is considered to ease the situation for transition crossing in the SPS by halving the number of ions in the same longitudinal emittance, so the number of decay ring merging steps is unaffected.

Finally, h=21 results in a bunch spacing such that the 40 MHz rf system employed at low energy in the SPS can hand over to the existing 200 MHz narrow-band one near transition in that machine. In an oversight in [1], it was considered that (in addition to relieving space charge) the new 40 MHz system would overcome all the frequency limitations of the 200 MHz one. But, with h=8 in the PS, it cannot do this at SPS injection and permit the rebucketing later in the cycle.

Emittance

The starting-point for estimates of emittance is the well-established high-intensity proton beam for fixed-target physics, since this essentially fills the aperture of the PS machine at injection. The normalized rms emittances of $\varepsilon^*_{H,V}$=15, 8 μm for protons at $\beta\gamma$=2.26 yield reference values of $\varepsilon^*_{H,V}$(ref, PSinj)=7.8, 4.2 μm scaling to a magnetic rigidity of 11 Tm (cf., 8 Tm in [1]) for ^{6}He ions. Then, ignoring any blow-up, the physical emittance is simply

$$\varepsilon(\text{ion},\text{machine}) = \frac{\varepsilon^*(\text{ref},\text{PSinj})}{\beta\gamma(\text{ion},\text{machine})} \qquad (1)$$

We assume the ^{18}Ne has the same normalized emittances as the ^{6}He because it comes from the linac with identical $\beta\gamma$ and is multi-turn injected into the RCS with the same geometrical set-up. This gives the following physical rms emittances according to Equation 1:

[μm]	^{6}He	^{18}Ne
RCS inj	16.4, 8.8	16.4, 8.8
PS inj	6.6, 3.5	4.0, 2.1
SPS inj	0.8, 0.4	0.5, 0.3

Table 1: $\varepsilon_{H,V}$ at injection in the three circular accelerators.

Tune shift

Considering for simplicity a round Gaussian beam of fully stripped ions, the self-field incoherent ("Laslett") tune shift is

$$\Delta Q = -\frac{Z^2}{A_p}\frac{3r_p}{4(\beta\gamma)^3 c}\frac{R}{\tau_b}\frac{N_b}{\varepsilon} \qquad (2)$$

where Z is the atomic number of the ion, A_p is the the ion-to-proton mass ratio, r_p is the classical proton radius, c is the speed of light, R is the mean radius of the machine, and N_b is the number of ions per bunch of duration τ_b and physical rms emittance ε.

We assume that τ_b is 80% of the bucket duration at injection in each machine and, for the purposes of illustration, take N_b for each ion species such that the integrated flux emanating from one straight section of the decay ring is 10^{18} yr^{-1}. This gives the following vertical (using ε_V since this is smaller than ε_H) tune shifts according to Table 1 and Equation 2:

	^{6}He	^{18}Ne
RCS inj	-0.022	-0.14
PS inj	-0.13	-0.36
SPS inj	-0.10	-0.28

Table 2: ΔQ_V at injection in the three circular accelerators.

In the case of the SPS, the calculations are based on the bunch that has the least time to decay in the PS before it is delivered into the 40 MHz (h=924) rf system.

For comparison, Equation 2 yields tune shifts of -0.34 and -0.078 for the so-called "ultimate" LHC proton beam at PS and SPS injection, respectively. Tune shifts exceeding 0.2 in magnitude have been tolerated at injection in the SPS [4].

"Bottom-up" results

The modified (with respect to [1]) parameters given in the Appendix to this note result in the following numbers of ions at each stage of the CERN Beta-beam facility. The decay losses are properly accounted for, but the transfer efficiencies between the different machines are assumed to be 100% except for the multi-turn injection into the RCS. The source rate is given (in atoms/s) at the entrance to the ECR, while the decay ring figure is the maximum number of stored ions immediately after injection from the SPS.

^{6}He:

Source rate	$2.\circ 10^{13}$
ECR	$1.87\circ 10^{12}$
RCS inj	$9.32\circ 10^{11}$
RCS	$9.03\circ 10^{11}$
PS inj	$1.12\circ 10^{13}$
PS	$9.58\circ 10^{12}$
SPS	$9.05\circ 10^{12}$
Decay Ring	$9.71\circ 10^{13}$

^{18}Ne:

Source rate	$8.\circ 10^{11}$
ECR	$2.29\circ 10^{10}$
RCS inj	$1.14\circ 10^{10}$
RCS	$1.13\circ 10^{10}$
PS inj	$1.9\circ 10^{11}$
PS	$1.81\circ 10^{11}$
SPS	$1.79\circ 10^{11}$
Decay Ring	$3.11\circ 10^{12}$

This corresponds to a total integrated flux in a ten-year run (5 years of ^{6}He plus 5 years of ^{18}Ne) of 1.46×10^{19} antineutrinos and 2.31×10^{17} neutrinos.

Conclusions

By raising the top energy of the RCS and by relaxing the constraint on total bunch length in the decay ring, space charge detuning proves not to be a complete show-stopper. Although it is already very much at the limit, one could imagine deliberately blowing up the emittance to improve the situation in the downstream machine.

The integrated fluxes achievable are higher in the new scenario, but the results for neutrinos coming from the decay of ^{18}Ne still do not match the expected order of magnitude. The figure for antineutrinos from ^{6}He has now reached 2.9×10^{18} yr^{-1}, but no safety margin is provided yet.

References

[1] M. Benedikt, S. Hancock and M. Lindroos, "The EURISOL Beta-beam facility parameter and intensity values, version 1, April 2005", EURISOL DS/TASK12/TN-05-01 (2005).

[2] M. Lindroos, “Possible ways of increasing the number of (anti-)neutrinos from the EURISOL Beta-beam facility”, EURISOL DS/TASK12/TN-05-02 (2005).

[3] M. Benedikt and S. Hancock, “A novel scheme for the injection and stacking of radioactive ions at high energy”, accepted for publication in NIM.

[4] H. Burkhardt et al., “Investigation of space charge effects and intrabeam scattering for lead ions in the SPS”, in Proc. European Particle Accelerator Conf., Lucerne (2004).

Appendix

The Mathematica package published in [1] has been modified:

```
Off[General::spell1];

c = 299792458;

(* All energies are in eV; "Tpern" = kinetic energy per nucleon. *)
setmachines := (
  bp = 1.2; (* 1.2s basic period. *)
  quantize[t_] := bp Ceiling[t/bp];
  ecrejTpern := 50 10^3 q/nnumber;
  ecraccumulationtime := rcscycletime - ecrdeadtime;
  ecrdeadtime = 2.5 10^-3;
  rcsinjTpern = 100 10^6;
  rcsefficiency = 0.5; (* Multi-turn injection. *)
  bunchingtime = 5 10^-3;
  rcsaccelerationtime := (rcscycletime - bunchingtime)/2;
  rcsBrhomax = 11; (* 500MeV/nucleon 6He equivalent. *)
  rcscycletime = 1/10;
  rcsbatches = 20;
  psinjTpern := Sqrt[(rcsBrhomax c q/nnumber)^2 + Epern^2] - Epern;
  psaccumulationtime := (rcsbatches-1) rcscycletime;
  psaccelerationtime = 0.8;
  psBrhomax = 86.7;
  pscycletime := quantize[psaccumulationtime + 2.0 psaccelerationtime
+ 0.1];
  psbatches = 1;
  spsinjTpern := Sqrt[(psBrhomax c q/nnumber)^2 + Epern^2] - Epern;
  spsaccumulationtime := (psbatches-1) pscycletime;
  spsaccelerationtime := (topTpern - spsinjTpern) *
    nnumber/(q 100 10^9); (* 100GeV/s proton equivalent. *)
  topgamma = 100;
  topTpern := Epern (topgamma - 1);
  spscycletime := quantize[spsaccumulationtime + 1.5
spsaccelerationtime + 1.0];
  spsrepetitiontime := Max[spscycletime, psbatches pscycletime];
  straightfraction = 0.36;
)

set6He := (
  name = "6He"; nnumber = 6; q = 2;
  Erest = 5.606 10^9; (* Rest mass in eV/c^2. *)
  Epern := Erest/nnumber;
  thalf = 0.81; (* Half-life at rest. *)
  sourcerate = 2 10^13; (* Effective rate, includes 40% efficiency. *)
  ecrefficiency = 1;
  setmachines;
  mergesratio = 15;
)

set18Ne := (
  name = "18Ne"; nnumber = 18; q = 10;
  Erest = 16.767 10^9; (* Rest mass in eV/c^2. *)
  Epern := Erest/nnumber;
  thalf = 1.67; (* Half-life at rest. *)
  sourcerate = 8 10^11; (* Effective rate, includes 40% efficiency. *)
  ecrefficiency = 0.3; (* Single charge state. *)
  setmachines;
```

```
  mergesratio = 20;
)

ecraccumulation := (ClearAll[n];
  gamma[t_] := 1 + ecrejTpern / Epern;
  decayrate[t_] := Log[2] n[t] / (gamma[t] thalf);
  eqns = {n'[t] == sourcerate - decayrate[t], n[0]==0};
  n[t_] = ecrefficiency n[t] /. DSolve[eqns, n[t], t] //First;
  nout0 = n[ecraccumulationtime]
)

rcsbunching := (ClearAll[n];
  gamma[t_] := 1 + rcsinjTpern / Epern;
  decayrate[t_] := Log[2] n[t] / (gamma[t] thalf);
  eqns = {n'[t] == -decayrate[t], n[0]==nout0};
  n[t_] = rcsefficiency n[t] /. DSolve[eqns, n[t], t] //First;
  nout1 = n[bunchingtime]
)

rcsacceleration := (ClearAll[n];
  rcsTpern[t_] := rcsinjTpern +
    (psinjTpern - rcsinjTpern) (1 - Cos[Pi t/rcsaccelerationtime])/2;
  gamma[t_] := 1 + rcsTpern[t] / Epern;
  decayrate[t_] := Log[2] n[t] / (gamma[t] thalf);
  eqns = {n'[t] == -decayrate[t], n[0]==nout1};
  n[t_] = n[t] /. DSolve[eqns, n[t], t] //First;
  nout2 = Simplify[ n[rcsaccelerationtime] ]
)

rcsout := (ecraccumulation; rcsbunching; rcsacceleration)

psaccumulation := (ClearAll[n];
  gamma[t_] := 1 + psinjTpern / Epern;
  decayrate[t_] := Log[2] n[t] / (gamma[t] thalf);
  eqns = {n'[t] == -decayrate[t], n[0]==nout2};
  nsinglebatch[t_] = n[t] /. DSolve[eqns, n[t], t] //First;
  n[t_] = Sum[UnitStep[t-t0] nsinglebatch[t-t0],
    {t0, 0,psaccumulationtime,rcscycletime}];
  nout3 = n[psaccumulationtime]
)

psacceleration := (ClearAll[n];
  psTpern[t_] := psinjTpern +
    (spsinjTpern - psinjTpern) t/psaccelerationtime;
  gamma[t_] := 1 + psTpern[t] / Epern;
  decayrate[t_] := Log[2] n[t] / (gamma[t] thalf);
  eqns = {n'[t] == -decayrate[t], n[0]==nout3};
  n[t_] = n[t] /. DSolve[eqns, n[t], t] //First;
  nout4 = n[psaccelerationtime]
)

psout := (rcsout; psaccumulation; psacceleration)

spsaccumulation := (ClearAll[n];
  gamma[t_] := 1 + spsinjTpern / Epern;
  decayrate[t_] := Log[2] n[t] / (gamma[t] thalf);
  eqns = {n'[t] == -decayrate[t], n[0]==nout4};
  nsinglebatch[t_] = n[t] /. DSolve[eqns, n[t], t] //First;
  n[t_] = Sum[UnitStep[t-t0] nsinglebatch[t-t0],
    {t0, 0,spsaccumulationtime,pscycletime}];
  nout5 = n[spsaccumulationtime]
```

```
)

spsacceleration := (ClearAll[n];
  spsTpern[t_] := spsinjTpern +
    (topTpern - spsinjTpern) t/spsaccelerationtime;
  gamma[t_] := 1 + spsTpern[t] / Epern;
  decayrate[t_] := Log[2] n[t] / (gamma[t] thalf);
  eqns = {n'[t] == -decayrate[t], n[0]==nout5};
  n[t_] = n[t] /. DSolve[eqns, n[t], t] //First;
  nout6 = n[spsaccelerationtime]
)

spsout := (psout; spsaccumulation; spsacceleration)

decayringaccumulation := (ClearAll[n];
  gamma[t_] := 1 + topTpern / Epern;
  decayrate[t_] := Log[2] n[t] / (gamma[t] thalf);
  eqns = {n'[t] == -decayrate[t], n[0]==nout6};
  nsinglebatch[t_] = n[t] /. DSolve[eqns, n[t], t] //First;
  n[t_] = Sum[UnitStep[t-t0]
UnitStep[t0+mergesratio*spsrepetitiontime-t] *
    nsinglebatch[t-t0], {t0, 0,t,spsrepetitiontime}];
  nout7 = n[(mergesratio-1)spsrepetitiontime]
)

fullchain := (spsout; decayringaccumulation)

annualrate := (1 - 2^-(mergesratio spsrepetitiontime/(topgamma
thalf))) *
  spsout straightfraction 10^7/spsrepetitiontime
```